Authors: Dmitrii Kofanov, Vladimir Kozlov, Alexander Libman, Nikita Zakharov
Date: 21.08.2022


use "Covid_Russia_Survey_DATASET.dta" 

***cd "your folder"


****** Eduaction groups (educ)
*1 - primary (8 grades), 
*2 - secondary education/school
*3 - techinical college
*4 - vocational high education
*5 - unfinished higher eduaction (6 semesters)
*6 - higher education (one or more): university degree
 
****** Income groups (wealth)
*1 - not enough money for basic needs 
*2 - basic needs covered but no money for cloth purchases
*3 - basic needs and cloth, but cannot afford household equipment
*4 - can afford anything but the car
*5 - can afford to buy a car but not unlimited resources
*6 - can afford whatever they want 
*98 - unidentified/no answer, 

****** Employment groups (job)
*1 - enterprenuer/work for themselves
*2 - manager, CEO
*3 - specialist without managing responsibilities
*4 - service person (no special training)
*5 - worker
*6 - student
*7 - pensioneer (by age)
*8 - pensioneer (by disability)
*9 - taking care of the household or parental leave
*10 - unemployed, looking for job
*11 - unemployed, not looking for job
*99 - refused to answer

****** City type (city)
*1 - Moscow
*2 - city with population over 500 thousand
*3 - city with population below 500 thousand but over 100 thousand
*4 - city with population below 100 thousand
*5 - village/town (no city status)

****** Gender
*1 - male
*2 - female
 


***TABLE E1: Trust in COVID-19 statistics in July after publication of official all-cause mortality (the effect of under-reporting on trust)


eststo clear
eststo:  reg  trust1 c.exc_may_rate excess_may_zero   , cl(id_reg)
eststo:  reg  trust1 c.exc_may_rate##c.edu_uni excess_may_zero   , cl(id_reg)
eststo:  reg  trust1 c.exc_may_rate##c.edu_uni excess_may_zero excess_m4_m6 c.gender age i.job i.wealth  i.educ i.city   , cl(id_reg)

esttab  using "TableE1_trust.txt",   nonotes ///
	nogap star(* 0.1 ** 0.05 *** 0.01) replace nocon b(3) se(3)    l /// 
	keep(excess_may_zero exc_may_rate c.exc_may_rate#c.edu_uni edu_uni ) o(excess_may_zero exc_may_rate c.exc_may_rate#c.edu_uni edu_uni )  ind("Controls = age") 	///
	stats(N r2 , fmt(0 2 ) )     ///
	ti("Table E1: Exposed under-reporting for May and trust in COVID-19 statistics, linear probability model") ///
	addnotes("Notes: Standard errors clustered at regional level in parentheses, * p < 0.1, ** p < 0.05, *** p < 0.01")  
 
   
 
 
grstyle clear
set scheme s2color
grstyle init
grstyle set plain, box
grstyle color background white
grstyle set color Set1
grstyle numstyle legend_cols 1
grstyle linestyle legend none
grstyle set legend 6 
grstyle color ci_area gs12%50		
		
		
***FIGURE 5: Trust in COVID-19 statistics	

	 
graph twoway (histogram trust_gap1 if   excess_may_zero==1,   frac color(none) lcolor(red%70)  discrete  width(.25 )  ) ///
			 (histogram trust_gap3  if  underreporting_group==1,   frac color(red%45) lcolor(red%70) discrete  width(.25 )   ) ///	
			 (histogram trust_gap2 if   underreporting_group==2,  frac color(red%75) lcolor(red%70) discrete width(.25 ) ) ///
			 ,  xtic(.5/4.5) xla(, tlc(none)) xlabel(4 "Fully yes" 3 "Rather yes" 2 "Only somewhat" 1 "Fully no" , labgap(.5) ) /// 
			 legend(order(3 "Regions with excess mortality in May: Under-reporting ABOVE the median" 2 "Regions with excess mortality in May: Under-reporting BELOW the median" 1 "Regions with no excess mortality in May")   size(small)  ) ///
			title("{bf:Question: Do you trust official COVID-19 statistics?}" "{sub:Full sample (N=1570)}" , height(7) )	ysize(5) ytitle("Fraction", size(small)   height(4)) scale(.85)	///
			name(trust_all, replace)  xline(2.5 , lcolor(black%55) lpattern(dash) )
			
graph export "Figure5_trust.png", as(png) replace width(4000)		

		
		 graph drop  trust_all 
		 
		 
			 
***FIGURE 6: Trust in COVID-19 statistics in respondents with and without university degreee

graph twoway (histogram trust_gap1 if   excess_may_zero==1 & edu_uni==1,   frac color(none) lcolor(orange%70)  discrete  width(.25 )  ) ///
			 (histogram trust_gap3  if  underreporting_group==1& edu_uni==1,   frac color(orange%45) lcolor(orange%70) discrete  width(.25 )   ) ///	
			 (histogram trust_gap2 if   underreporting_group==2& edu_uni==1,  frac color(orange%75) lcolor(orange%70) discrete width(.25 ) ) ///
			 ,  xtic(.5/4.5) xla(, tlc(none)) xlabel(4 "Fully yes" 3 "Rather yes" 2 "Only somewhat" 1 "Fully no" , labgap(.5) ) /// 
			 legend(order(3 "Regions with excess mortality in May: Under-reporting ABOVE the median" 2 "Regions with excess mortality in May: Under-reporting BELOW the median" 1 "Regions with no excess mortality in May" )  ) ///
			 title("{bf:Question: Do you trust official COVID-19 statistics?}" "{sub:Subsample: University degree (n=671)}", height(7) ) ytitle("Fraction", size(small)   height(4)) 	///
			 name(trust_edu, replace) 		 xline(2.5 , lcolor(black%55) lpattern(dash) ) 

graph twoway (histogram trust_gap1 if   excess_may_zero==1 & edu_uni==0,   frac color(none) lcolor(green%70)  discrete  width(.25 )  ) ///
			 (histogram trust_gap3  if  underreporting_group==1& edu_uni==0,   frac color(green%45) lcolor(green%70) discrete  width(.25 )   ) ///	
			 (histogram trust_gap2 if   underreporting_group==2& edu_uni==0,  frac color(green%75) lcolor(green%70) discrete width(.25 ) ) ///
			 ,  xtic(.5/4.5) xla(, tlc(none)) xlabel(4 "Fully yes" 3 "Rather yes" 2 "Only somewhat" 1 "Fully no" , labgap(.5) ) /// 
			 legend(order(3 "Regions with excess mortality in May: Under-reporting ABOVE the median" 2 "Regions with excess mortality in May: Under-reporting BELOW the median" 1 "Regions with no excess mortality in May" )  ) ///
			 title("{bf:Question: Do you trust official COVID-19 statistics?}" "{sub:Subsample: No university degree (n=899)}" , height(7) )	ytitle("Fraction", size(small)   height(4)) 		///
			 name(trust_noedu, replace) 	 xline(2.5 , lcolor(black%55) lpattern(dash) ) 

			 
graph combine      trust_edu trust_noedu, col(2)  iscale(.55)    ysize(3)  xsize(5)   imargin(2 0 0 0) ycommon xcommon  

graph export "Figure6_trust_education.png", as(png) replace width(4000)		
		  
		 graph drop     trust_edu trust_noedu,
			 		 
			 
 
